
Spike2 and Signal Training Days

Our US Spike2 and Signal training courses will be held on Thursday 1* and Friday 2" of 3
November, prior to the Society for Neuroscience meeting, in San Diego. Venue S
confirmation and booking forms will be released shortly. If you would like to be kept up-to-

date with the latest training day information, please contact us.

Meetings and events £

Life Sciences 2007

Scotland Exhibition and Convention Centre
Glasgow

July 8" — 12" 2007

IBRO World Congress of Neuroscience
Melbourne Exhibition and Convention Centre
Melbourne, Australia

July 12" — 16™ 2007

IBRO Satellite Meeting: Motor Control at the Top End
Darwin, Australia
July 18" — 21% 2007

8th International Congress of Neuroethology
University of British Columbia

Vancouver, Canada

July 22" — 27" 2007

Latest versions of Spike2 and Signal

Updates for Spike2 and Signal are available from the CED downloads page, or by
clicking on the links in the table below. Demonstration versions of the latest software are
also available.

Spike2 downloads Signal downloads
Spike?2 version 6.04 Signal version 3.09
Spike?2 version 5.17 Signal version 2.16
Spike2 demo Signal demo

New features available in Spike2 version 6.04 include:
e New HTML help system (supported by Vista)
e Clustering dialogs now update with incoming spike data on-line
e New icons for Spike2 files

A full list of the new features and changes can be found on the website and in the on-line
help.


http://www.lifesciences2007.org/
http://www.ibro2007.org/
http://www.sapmea.asn.au/conventions/motor2007/index.html
http://www.confmanager.com/main.cfm?cid=720
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/sigw2u.shtml
http://www.ced.co.uk/sigdemu.shtml

I have just started using Spike2 and would like to know how | can automatically
detect peaks and other features of interest in my recorded ECG trace.

Spike2 can search for and detect features in waveform data and step to time-
stamp markers and events using the built-in Active Cursor functions. Search
methods include, but are not limited to:

e Peaks and troughs

e Threshold crossings, data within and data outside thresholds
e Slope measurements

e Percentage repolarisation

Up to ten cursors (0 — 9) are available in a data file view and individual cursors
can search any channel for features of interest. Cursor 0 acts as the reference or
iterator cursor so that, when it searches for new features, all other active cursor
positions are updated in response. For this reason Cursor 0 is nearly always
used to find a common reference around which other features can be found by
additional active cursors. Cursors can be created and Active modes set from
within the Cursor menu.

A simple example: finding peaks

One of the most common requirements when analysing an ECG signal is to be
able to detect and mark the R wave of successive beats. This could then be used
to determine heart rate or R-R interval times. In the example below, Cursor O is
set up to search our ECG data channel for Peaks that have an amplitude change
of 4mV. The amplitude value is a measure of the level that the waveform signal
falls after a peak to be classified, which allows accurate detection of peaks on a
signal with a baseline drift.

Cursor 0 settings for R-wave peak detection

You can check the behaviour of active cursors by holding down Ctrl + Shift
and using the left and right arrow cursor keys. This steps Cursor 0 between
features that match the search method parameters.

Finding additional features:

Back to top
Now that Cursor 0 is set to find the R-wave peaks, we can set additional active

cursors to find other features, in this case the Q and S components of the QRS
complex. Cursor 1 is set to find the S wave by using the Trough search method
with an amplitude level of 1mV (For trough search methods, the amplitude value
is a measure of the level that the waveform rises after a trough). Cursor 1 is then
set to search for the trough in the time range between Cursor(0) and
Cursor(0)+1 second.

Cursor 1 set for S-wave detection

The Q wave can be found by using Cursor 2 to search for the turning point
(where the waveform slope changes sign) in the short time range prior to the R-
wave. In our example, a start time of Cursor(0)-0.05 seconds and end time of
Cursor(0)-0.02 seconds finds the Q wave in the QRS complex. You can
define your own labels for cursors as displayed below, details can be found in
eNewsletter 26.

Cursor 2 set for Q wave with user-defined cursor labels

Multiple results from cursor positions can be plotted or marked automatically to
an XY view or data channel in the file using the Measurement functions available
from the Analysis menu. For more information see the Spike2 on-line help or see
our feature on taking measurements in the next eNewsletter.

Signal

I will shortly be setting up a system for recording responses from trans-cranial
magnetic stimulation, and would like to know how Signal could be used for
triggering/controlling TMS units

Signal can generate digital and analog outputs to trigger current, voltage and
magnetic stimulators while simultaneously recording and analyzing responses.
From the Pulse Outputs section in the sampling configuration you can quickly
set-up pulses with a drag-and-drop interface and use multiple states to create a
series of pulse protocols for stimulator timing and control. A simple example
would be to use a single pulse output to trigger an electrical stimulator at the start
of a sampling sweep, and record the response.

Signal also includes configuration support for the MagStim range of magnetic
stimulators when sampling. This allows direct control of MagStim settings via the
serial line in addition to using the standard pulse outputs for timing.



The MagStim controls are accessed from the States tab in the sampling

configuration. This provides a dialog from which you can select the MagStim type,

communication port and independent stimulation settings, such as intensity,
pulse interval etc. for each state, based on the type of MagStim used.

Configuration settings for a MagStim BiStim unit

Detailed explanations for the MagStim configurations can be found in the Signal
on-line help.

Scripts: Spike2 m

Q.

I record and analyse responses to repetitive electrical stimulation and was
wondering if there was any way to number the stimulus markers in a data file?
This would be useful when visually examining the file as | would know which
stimulus number corresponded to each response.

The attached script, StimNumber.s2s, can be used to give an identifying
number to stimulus events, spike shapes or markers. The script creates a new
TextMark memory channel that holds text strings starting at 1 for the time of the
first event in range.

4 Stimulus 1+

Start End
m1Memory 12 3 4 5 6 7 8 9 10 -1 12 13 14

5.8 8.9 9.0 9.1 9.2 9.3 94
s

Channel m1 displays the corresponding stimulus numbers

Scripts: Signal E\
Q. | would like to be able to only view frames of a given state when paging through a

A.

data file.

The attached script, StateFilter.sgs, allows the user to select which state
will be displayed and can then step through the frames of that state using the
associated script toolbar buttons.

Did you know...?

Included in the main Spike2 folder is an application called s2video which will record video
and audio data alongside the sampled data file. Video and audio data is streamed directly
to disk as soon as sampling starts, time-locking it to the data file. The s2video application
is based on Microsoft DirectShow, which is part of DirectX and to run multimedia capture

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
seripts in this newsletter
please let us know.

Right-click the script icon
and save to disk.

If you have any problems
opening the embedded
scripts in this newsletter
please let us know.

'$StimNumber.s2s|Script to generate textmark channel to number stimuli or other events

'within a specified time range





var stCh%;              'Stimulus channel

var nCh%;               'Number channel

var sTime,eTime;        'Time range

var data%;              'Data file handle

var getRange%;          

var stims%;             'Number of events in time range

var i%;                 'Counter

var code%[4];           'Code array for textmarks

var stim$;              'String to print to textmarks

var stTime;             'Event time



data% := FrontView();               'Check if current view is a data file

if ViewKind(data%) <> 0 then

    Message("Current view is not a time view\nPlease open a data file");    

    data%:= FileOpen("",0,1);       'Prompt user to open file if not

    if data% < 0 then

        Message("Unable to open a data file!");

    Halt;endif;                 'If unable to open a file, halt script

endif;



DlgCreate("Stimulus numbering");            'Dialog to set channel

DlgChan(1, "Select stimulus channel",62);

DlgShow(stCh%);



CursorSet(2);                               'Allow user to specify whole file of cursor defined region

CursorLabel(4,1,"Start");

CursorLabel(4,2,"End");

getRange% := Interact("Select time range for stimulus numbering",1023,0,"Whole file","Cursors");

docase

case getRange% = 1 then         'If whole file selected

    sTime := 0;

    eTime := MaxTime();

case getRange% = 2 then         'If cursor positions used

    sTime := Cursor(1);

    eTime := Cursor(2);

endcase;



stims% := Count(stCh%,sTime,eTime);         'Count events

if stims% > 0 then                          'If any events in time range

    stTime := sTime;                        

    nCh% := MemChan(8,6);                   'Create memory channel

    for i% := 1 to stims% do                'Steps through each event

        stTime := NextTime(stCh%,stTime);   'Get time

        MemSetItem(nCh%,0,stTime,code%[],Str$(i%));     'Add textmark with event number

        next;

    DrawMode(nCh%,14,2);                    'Draw in text mode

    ChanShow(nCh%);                         'Show channel

endif;





StimNumber.s2s


'$StateFilter|Script to only display frames of a selected state number



var dataFile%;		'Handle of data file

var numFrames%;		'Number of frames

var state%;         'Selected state

var stateNo$;       'Text display for toolbar

var index%;         'Index counter for current frame

var maxFrames%;     'Max. no. of frames of state n

var ok%;            ' Check for dialog entry    

var dispFrames%[1000];  'Array to hold all frames

var maxState%;      'Current state count

var cFrame%;        'No. of current frame

var cFrame$;        'Text of current frame no.



dataFile% := FrontView();		'Show an open data file



if ViewKind(dataFile%) <> 0 then

	Message("No Signal data file open\nPlease select a data file");

	dataFile%:= FileOpen("",0,1);	'if no data file found, prompt user to open one

	

	if dataFile% < 0 then				'If user cancels or error occurs

   		Message("Unable to open a file!\nRun script to try again");		'Warn user 		

		Halt;		'and quit script	

	endif;

endif;

Window(0,0,100,100);	'Make the data full screen



numFrames%:= FrameCount();		'Get no. of frames

GetFrameStates%(dispFrames%[]);		'Get frame states

DoToolbar%();





Func GetFrameStates%(dispFrames%[]);

var c%;                 'Counter for frames

var d%:=1;              'Counter for states

var currState%;         'State of current frame



for c%:=1 to numFrames% do		'Cycle through frames

	

	currState%:= FrameState(c%);	'get current state

	

	if currState% > maxState% then

		maxState%:= currState%;		'If state number greater than previous, set to max state

	endif;

next;



	DlgCreate("State Filter");

	DlgInteger(1,"State to display",1,maxState%);		'User selects state to display

	ok%:= DlgShow(state%);



	if ok% < 0 then

		Message("Script cancelled");

	Halt;endif;



	stateNo$:= Print$("Display of State %d",state%);	'Toolbar display

    

for c%:=1 to numFrames% do      'Cycle through frames

	if FrameState(c%) = state% then     'And match state to state filter

		dispFrames%[d%]:= c%;

		d%+=1;                  'Keep count of frames that match

	endif;

next;



dispFrames%[0]:= d%-1;

maxFrames%:= d%-1;            'Get count of frames that match

Frame(dispFrames%[1]);      'Display the first frame of state n

cFrame%:= Frame();          'Get current frame no.

cFrame$:= Print$(": Frame %d",cFrame%);     'Print for toolbar display

index%:= 1;

return 1;

end;



Func DoToolbar%()           'Toolbar to allow control of frame display

ToolbarClear();

ToolbarSet(1,"&Quit");      

ToolbarSet(3,">>",Fwd%);

ToolbarSet(4,"<<",Rew%);

return Toolbar(stateNo$+cFrame$,1023);  'Display current frame and state

end;



Func Fwd%()

if index% < maxFrames% then;        'If current display is not the last frame in the list

	index% += 1;                    'Increment index

	Frame(dispFrames%[index%]);     'Display the next frame

    cFrame%:= Frame();              'Get current frame no.

    cFrame$:= Print$(": Frame %d",cFrame%);     

    ToolbarText(stateNo$+cFrame$);      'Update the toolbar display

endif;

return 1;

end;



Func Rew%()

if index% > 1 then                  'If current display is not the first frame in the list

	index% -= 1;                    'Decrement index

	Frame(dispFrames%[index%]);     'Display the previous frame

    cFrame%:= Frame();              'Get current frame no.

    cFrame$:= Print$(": Frame %d",cFrame%);

    ToolbarText(stateNo$+cFrame$);      'Update the toolbar display

endif;

return 1;

end;



StateFilter.sgs


alongside sampling you need a suitably powerful computer. See the Spike2 on-line help
for details of system requirements.

Recent questions Back to top

Q. | have a series of files with the data stored as text from old experiments that |
would like to review and analyse. Can | read these files using Spike2 or Signal?

A. Both Spike2 and Signal can import text files as well as a range of 3" party data
files stored in other formats for analysis. The File menu Import function opens a
dialog from which you can select the type of file to import from the drop-down list.
If you wish to import data files in a format that is not currently supported, please
contact us.

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk.

All trademarks are acknowledged to be the trademarks of the registered holders.

